P5

5. PRAKTIKA- KURBEN ADIERAZPENA KOORDENATU POLARRETAN

▼ Proposatutako Ariketa P-5.1

Irudikatu ondorengo zirkuluen familia:



▼ Soluzioa P-5.1

Zirkunferentziaren ekuazio orokorra zentrua (a,b) eta erradioa c izanik

```
Clear["Global`*"]
ek = (x - a)^2 + (y - b)^2 == c^2
```

$(-a+x)^2 + (-b+y)^2 == c^2$

1. Zirkulua: Zentrua OY ardatzean duena, (a,b)=(0,b), a=0 eta c=b izanik

```
ek1 = ek /. {a → 0, c → b}
```

$x^2 + (-b+y)^2 == b^2$

```
polar1 = ek1 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$r[t]^2 == 2\, b\, r[t]\, Sin[t]$

```
r[t]^2 == 2 b r[t] Sin[t]
```

$r[t]^2 == 2\, b\, r[t]\, Sin[t]$

```
Solve[polar1, r[t]]
```

{{r[t] → 0}, {r[t] → 2 b Sin[t]}}

```
zirkulu1[t_, b_] = 2 * b Sin[t];
```

2. Zirkulua: Zentrua OX ardatzean duena, (a,b)=(a,0), b=0 eta c=a izanik

```
ek2 = ek /. {b → 0, c → a}
```

$(-a + x)^2 + y^2 == a^2$

```
polar2 = ek2 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$2\, a\, Cos[t]\, r[t] == r[t]^2$

```
Solve[polar2, r[t]]
```

{{r[t] → 0}, {r[t] → 2 a Cos[t]}}

```
zirkulu2[t_, a_] = 2 * a Cos[t];
```

3. Zirkulua: Zentrua jatorrian duena, (a,b)=(0,0), a=0 eta b=0 izanik

```
ek3 = ek /. {a → 0, b → 0}
```

$x^2 + y^2 == c^2$

```
polar3 = ek3 /. {x → r[t] * Cos[t], y → r[t] * Sin[t]} // Simplify
```

$c^2 == r[t]^2$

```
Solve[polar3, r[t]]
```

{{r[t] → -c}, {r[t] → c}}

```
zirkulu3[t_, a_] = a;
```

Zirkuluen familia

```
zir1 = PolarPlot[Evaluate[Table[zirkulu1[t, b], {b, 1, 5}]],
   {t, 0, 2 * π}, PlotStyle → RGBColor[1, 0, 0.2], PlotLabel → r == 2 b Sin[t]];
zir2 = PolarPlot[Evaluate[Table[zirkulu2[t, a], {a, 1, 5}]], {t, 0, π},
   PlotStyle → RGBColor[0.5, 1, 0], PlotLabel → r == 2 a Cos[t]];
zir3 = PolarPlot[Evaluate[Table[zirkulu3[t, c], {c, 2, 10, 2}]],
   {t, 0, 2 * π}, PlotStyle → RGBColor[0.5, 0, 1], PlotLabel → r == c];
Show[zir1, zir2, zir3, PlotLabel → Style["ZIRKULUEN FAMILIA",
   Bold, 14, RGBColor[0.3, 0.2, 0.7]]]
```



▼ Proposatutako Ariketa P-5.2

Lortu ondorengo kardioideen familiaren grafikoa:



▼ Soluzioa P-5.2

1. Kardioidea

```
kardioide1[t_, a_] = a (1 + Cos[t]);
kar1 = PolarPlot[kardioide1[t, 1], {t, 0, 2 π}, PlotStyle → Red]
```



2. Kardioidea

```
kardioide2[t_, a_] = a (1 - Cos[t]);
kar2 = PolarPlot[kardioide2[t, 1], {t, 0, 2 π}, PlotStyle → Green]
```



3. Kardioidea

```
kardioide3[t_, a_] = a (1 + Sin[t]);
kar3 = PolarPlot[kardioide3[t, 1], {t, 0, 2 π}, PlotStyle → Purple]
```



4. Kardioidea

```
kardioide4[t_, a_] = a (1 - Sin[t]);
kar4 = PolarPlot[kardioide4[t, 1], {t, 0, 2 π}, PlotStyle → Orange]
```



Kardioideen familia

```
Show[kar1, kar2, kar3, kar4]
```



▾ Proposatutako Ariketa P-5.3

Kalkulatu ondorengo lemniskata famiaren grafikoa:



▼ Soluzioa P-5.3

```
lemniskata[t_, a_] = a (Cos[2 * t]) ^ (1 / 2);

PolarPlot[Evaluate[Table[lemniskata[t, a], {a, 2, 8, 2}]], {t, 0, 2 Pi},
 PlotLabel → Style["LEMNISKATA FAMILIA", Bold, 14], Background → LightGreen]
```



▼ Proposatutako Ariketa P-5.4

Lortu espiralen ondoko grafikoa:



▼ Soluzioa P-5.4

```
esparq[θ_, a_, b_, x_] = a + b θ^(1 / x);

PolarPlot[{esparq[θ, 0, 1, 1], esparq[θ, 0, -1, 1]},
 {θ, 0, 4 Pi}, PlotStyle → {Purple, Orange}]
```